

AVISS — Automated Verification of Infinite State Systems

(IST-2000-26410)

Deliverable D2.2

Specification of the Intermediate Format IF

1 Introduction

In deliverable D2.1 we specified the High Level Protocol Specification Language (HLPSTL) and hence the front-end of a compiler from HLPSTL to Intermediate Format (IF). In this deliverable we specify the IF and hence the back-end of the HLPSTL2IF compiler.

The output side of the compiler is prone to evolve according to the development of the tools that work on the IF. To be flexible in experiments, we first describe here a grammar that matches a super-set of the output of the HLPSTL2IF compiler: the generic IF. This grammar is the specification of the parser of the tools that work on the IF. Second, we give a grammar matching exactly the current output of HLPSTL2IF and a grammar that describes an experimental variant which excludes checking for type-flaws but allows faster search for security holes that are not type-flaws.

2 Notation

We will use throughout this document a standard notation for grammar, that is:

- $[xyz]$ is a “character class”. This pattern, for example, matches either x or y or z .
- $x - y$ stands for an interval in a character class (all the characters between x and y).
- $^{\wedge}$ is used for exclusion in a character class. For example, $[a - z^{\wedge}f - m]$ matches all the characters between a and z except those between f and m .
- $r?$, where r is a pattern, stands for zero or one occurrence of pattern r .
- $r+$, where r is a pattern, stands for one or more repetition of pattern r .
- r^* , where r is a pattern, stands for zero or more repetition of pattern r .
- EOL stands for an End-Of-Line character.

In order to distinguish between characters and non-terminal symbols, we write characters in small caps, e.g. `EXAMPLE`, and non-terminal symbols in italics, e.g. *example*.

In addition, we use the following shorthands:

<i>Variable</i>	::=	$[x]Text$
<i>ConstText</i>	::=	$[A-ZA-Z^{\wedge}x] Text$
<i>Time</i>	::=	$s(Time) \mid xTIME$
<i>Digit</i>	::=	$[0-9]$
<i>Number</i>	::=	$Digit+$
<i>Char</i>	::=	$[A-ZA-Z]$
<i>Text</i>	::=	$(Digit \mid Char \mid _)+$
<i>GenericText</i>	::=	$[ASCII(32) - ASCII(126)]^*$

GenericText matches any sequence of printable character that does not contain the EOL character.

3 Intermediate Format File Description

In addition to rewrite rules that describe the protocol, we have to add, in the intermediate format, information that describes to what part of the protocol description a rule is related to. This information can then be used to give a specific status to some rules. We formalize this by giving, for the Intermediate Format file, a grammar that describes how rules are introduced. The current grammar for this description is given in Figure 1.

The rest of this deliverable focusses on the description of the *Rule* item.

¹The *RuleIdentifier* parts of two different labelled rules are always different.

```

FileDescription ::= Type EOL ( LabelledRule | CommentLine | EOL)*
Type           ::= # OPTION=UNTYPED | # OPTION=TYPED
CommentLine    ::= ## GenericText EOL
LabelledRule   ::= Label Rule EOL
Label          ::= # LB=RuleIdentifier , TYPE=RuleSection EOL
RuleIdentifier  ::= Text1
RuleSection    ::= PROTOCOL_RULES | INVARIANT_RULES | DECOMPOSITION_RULES
                  | INTRUDER_RULES | INIT | GOAL |

```

Figure 1: Description of the file containing the Intermediate Format

4 Generic Intermediate Format

As mentioned above, it is foreseeable that, in the project months to come, there will be some changes in the IF. We thus give now a grammar that is supposed to be stable and that provides a framework for experiments:

```

Rule           ::= State => State | State
State          ::= Term | Term · Term
Term           ::= H(Time) | Principal | MessageTerm | IntrudersKnowledge | Secret | Message
Principal      ::= w(Message,Message,Message,Message,Message,Message,Message)
MessageTerm    ::= m(Message,Message,Message,Message,Message,Message)
IntrudersKnowledge ::= i(Message)
Secret         ::= SECRET(Message,F(Message))

```

According to this grammar, a rule is a transition from one state to a new state (except for rules that describe initial states and goal states). A state is a list of terms and a term is either *Principal* (expressing the state of one principal: his short- and long-term knowledge and the message he is expecting), *MessageTerm* (a message that is sent and not yet received), *IntrudersKnowledge* (what the intruder has already found out), or *Secret* (stating that some information must be kept secret).

And we define the messages with:

```

Message ::= CRYPT(Message,Message) | SCRYPT(Message,Message)
          | C(Message,Message) | FUNCT(Message,Message) | RCRYPT(Message,Message)
          | MR(Message) | NONCE(Message) | PK(Message) | SK(Message)
          | FU(Message) | Message' | ConstText | Number | Variable | Time

```

An atomic message like a key, a nonce, or a principal name is declared by a name of alphanumeric characters. It can be thought of as representing a sequence of bits in real world protocols. Type information is added to these atomic items like $MR(A)$, $SK(KAB)$, or $PK(KA)$. This type information determines which properties the Atom has, e.g. we assume

$$SCRYPT(SK(KAB), SCRYPT(SK(KAB), XMESSAGE)) = XMESSAGE.$$

A message can now be an atomic message or a composed message (e.g. coupling c , encryption $CRYPT$, etc.).

5 Current Intermediate Format

In the previous section we gave a generic framework for the IF. However there are for instance many message terms that don't make sense like $PK(SK(ka))$. In this section we give a tighter

description of the messages and rules that can occur in the output of HLP2IF. We first consider the (default) untyped version and describe the differences for the typed version in the next section.

The grammar for HLP2IF's output can be split in two parts. The first of these is about terms exchanged during the protocol, and the second one is about the “administration of these terms”, that is representation of messages, of principals, roles, intruder's knowledge and so on. First give the grammar for the messages:

```

Message ::= Binary | Unary
Binary  ::= Couple | SymmetricCypher | PublicCypher | Function | XOR
Couple   ::= c(Message,Message)
SymmetricCypher ::= SCRYPT(SK(Atomic),Message) | SCRYPT(Variable,Message)
PublicCypher  ::= CRYPT(PK(Atomic),Message) | CRYPT(Variable',Message)
               CRYPT(PK(Atomic)',Message) | CRYPT(Variable,Message)
Function      ::= FUNCT(FU(Atomic),Message) | FUNCT(Variable,Message)
XOR          ::= RCRYPT(Message,Message)
Unary        ::= Variable | MR(Atomic) | PK(Atomic) | PK(Atomic)' | SK(Atomic) |
               NONCE(Atomic) | FU(Atomic)
Atomic      ::= Const
Const       ::= FreshConst | ConstText
FreshConst  ::= c(ConstText,Time) | c(NI,NI)

```

Let us now consider the roles description. Formally, we represent *roles* when using the `--rules` option, and *principals* when using the `--init` option. And during a protocol execution, roles get naturally instantiated as principals. In order to do this, the terms representing principals and roles have exactly the same structure. The grammar is:

```

Principal ::= w(Step,Sender,Receiver,AcqKnowledge,InitialKnowledge,Boolean,Session)
Step      ::= Number | Variable
Sender    ::= Id
Receiver  ::= Id
Id        ::= MR(Atomic) | Variable
AcqKnowledge ::= ListOfMessages
InitialKnowledge ::= ListOfMessages
ListOfMessages ::= ETC | ETC2 | Variable | c(Message,ListOfMessages)
Boolean      ::= TRUE | FALSE | Variable
Session     ::= s(Session) | Number | Variable

```

The message description is based on the same idea:

```

MessageTerm ::= m(Step,RealSender,OfficialSender,Receiver,Contents,Session)
RealSender  ::= Id
OfficialSender ::= Id
Receiver    ::= Id
Contents    ::= Message

```

The Intruder's knowledge description is:

```

IntrudersKnowledge ::= i(Message)

```

A state of the protocol is given as a non-empty multiset of terms, in the following way:

```

State ::= Term | State · Term
Term  ::= Principal | MessageTerm | IntrudersKnowledge | Secret
Secret ::= SECRET(Variable,F(Session))

```

We can now give rewrite rules on these states:

<i>Rule</i>	$::=$	<i>Initial</i> <i>MessageRule</i> <i>IntruderRule</i> <i>FlawRule</i> <i>SimplificationRule</i>
<i>Initial</i>	$::=$	$H(x_{TIME}) \cdot State$
<i>MessageRule</i>	$::=$	<i>InitialMessageRule</i> <i>RegularMessageRule</i> <i>LastMessageRule</i>
<i>InitialMessageRule</i>	$::=$	$H(s(x_{TIME})) \cdot Principal =>$ $H(x_{TIME}) \cdot MessageTerm \cdot Principal \cdot Secret$
<i>RegularMessageRule</i>	$::=$	$H(s(x_{TIME})) \cdot MessageTerm \cdot Principal =>$ $H(x_{TIME}) \cdot MessageTerm \cdot Principal \cdot Secret$
<i>LastMessageRule</i>	$::=$	$H(s(x_{TIME})) \cdot MessageTerm \cdot Principal =>$ $H(x_{TIME}) \cdot Principal \cdot Secret$
<i>IntruderRule</i>	$::=$	<i>Memorize</i> <i>Divert</i> <i>EavesDropping</i> <i>Impersonate</i>
<i>Memorize</i>	$::=$	$H(s(x_{TIME})) \cdot M(x1, x2, x3, MR(1), x5, x6) =>$ $H(x_{TIME}) \cdot I(x2) \cdot I(x5)$
<i>Divert</i>	$::=$	$H(s(x_{TIME})) \cdot M(x1, MR(Const), x3, x4, x5, x6) =>$ $H(x_{TIME}) \cdot I(MR(Const)) \cdot I(x3) \cdot I(x4) \cdot I(x5)$
<i>EavesDropping</i>	$::=$	$H(s(x_{TIME})) \cdot M(x1, MR(Const), x3, x4, x5, x6) =>$ $H(x_{TIME}) \cdot M(x1, MR(Const), x3, x4, x5, x6) \cdot I(MR(Const)) \cdot I(x3) \cdot I(x4) \cdot I(x5)$
<i>Impersonate</i>	$::=$	$H(s(x_{TIME})) \cdot Principal \cdot (IntrudersKnowledge)^+ =>$ $H(x_{TIME}) \cdot MessageTerm \cdot Principal \cdot (IntrudersKnowledge)^+$
<i>FlawRule</i>	$::=$	<i>State</i>
<i>SimplificationRule</i>	$::=$	<i>State</i> $=>$ <i>State</i> <i>IdempotenceRule</i> <i>SpecialRule</i>
<i>IdempotenceRule</i>	$::=$	$x1 \cdot x1 => x1$
<i>SpecialRule</i>	$::=$	$F(s(x1)) => F(x1)$

6 The Typed Version

Note that in the above grammar variables are not typed, e.g. a term like $MR(x4)$ will never occur, only constants are typed like $MR(alice)$. While this partially permits to model type-flaws (mistaking a nonce for a symmetric key, for example), this leads to a branching explosion in the rules where the intruder composes a message (impersonate rules). As a consequence, we have defined a “typed” version of the Intermediate Format that restricts the domain of variables by specifying a type.

For example, suppose that the expected message is an unknown nonce Na . In the “untyped” version of the Intermediate Format, the rule that permits the intruder to compose a message is of the form:

$$PrincipalTerm \cdot I(xNa) => M(..., xNa, ...) \cdot PrincipalTerm \cdot I(xNa)$$

This rule can be used with various values for the variable xNa , including a symmetric key, or an un-decomposed cypher and so on. The number of possibilities is the number of pieces of knowledge the intruder has. On the other hand, when one keeps the type information, as in the “typed” version, this rule becomes:

$$PrincipalTerm \cdot I(NONCE(xNa)) => M(..., NONCE(xNa), ...) \cdot PrincipalTerm \cdot I(NONCE(xNa))$$

In this case, this rule can only be applied with a piece of knowledge of the intruder that is also “typed” with nonce, thus reducing the possibilities of application. The drawback is that it is impossible to find type flaws in this case. The fact that the output of HLP2IF is typed or not is stamped by the first line of the file in the Intermediate Format (see Figure 1). The rest of the file follows the same grammar as in the “untyped” version, except for the five following rules, which deal with the construction of message terms:

<i>SymmetricCypher</i>	$::=$	$SCRYPT(SK(Atomic), Message)$
<i>PublicCypher</i>	$::=$	$CRYPT(PK(Atomic), Message) \mid CRYPT(PK(Atomic)', Message)$
<i>Function</i>	$::=$	$FUNCT(FU(Atomic), Message)$
<i>Atomic</i>	$::=$	<i>Const</i> <i>Variable</i>
<i>Id</i>	$::=$	$MR(Atomic)$